
Moment matching for Asian Options

02-22-17

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
  StyleDefinitions -> Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

My last previous notebook on this topic was *Moment matching for Asian Options 01-22-11*, which consolidated several pieces of earlier work. As I worked through that notebook, I realized that advances in Mathematica technology suggested a rework of the old calculations.

Included are

- 1) Demonstration that arithmetic average price is approximated by log normal distribution
- 2) Moment matching for a continuously sampled average
- 3) Moment matching for a discretely sampled average

Background

Asian options are financial derivatives that depend on a time average of the price. The archetype is an option with a simple put or call payoff on the arithmetic average of periodically sampled prices.

$$A(T) = \frac{1}{n+1} \sum_{i=0}^n S(t_i) \tag{1}$$

where $T = n \Delta t$ and Δt is a constant interval of time, say one day.

For the example of a European style call option on this average, the fair value is given by

$$C(K, T, t) = \mathbb{E}_Q[e^{-r(T-t)} \max(A(T) - K, 0)] \quad (2)$$

an Asian call option with fixed strike. There are various Asian option types depending on how the price is sampled, how the average is constructed, special payoff features and more. However, central to the valuation of Asian options is knowledge is how the average $A(T)$ is distributed.

Within the Black Scholes framework, stock prices are assumed to be lognormally distributed. Thus, $A(T)$ is an arithmetic sum of lognormally distributed random variables, specifically

$$A(T) = \frac{1}{n+1} \left(S_0 + S_0 e^{\left(r - \frac{\sigma^2}{2}\right)\Delta t + \epsilon_1 \sigma \sqrt{\Delta t}} + \dots + S_0 e^{\left(r - \frac{\sigma^2}{2}\right)i\Delta t + (\epsilon_1 + \dots + \epsilon_i) \sigma \sqrt{i\Delta t}} + \dots \right) \quad (3)$$

This distribution for this sum has no closed form and must be approximated. It is observed that $A(T)$ is reasonably approximated by a lognormal distribution. If the parameters of the approximating distribution can be obtained, then the expectation in (2) can be evaluated in closed form, just like plain vanilla options.

The parameters of the approximating lognormal distribution for A are determined by matching the moments of the assumed lognormal distribution to the moments calculated from (3), a procedure called moment matching. Since a lognormal distribution is determined by two parameters, only two moments need be estimated.

In this notebook, I illustrate how *Mathematica* facilitates can be used to facilitate some of the calculations associated with moment matching. In my work as a quant, I generated many *Mathematica* notebooks on Asian options.

I The distribution $A(T)$ is approximated by a lognormal distribution

I start by using Monte Carlo simulation to illustrate how $A(T)$ is reasonably approximated by a lognormal distribution. The assumed stock dynamics are

$$\frac{dS_t}{S_t} = \mu dt + \sigma dz_t$$

$$S_t = S_0 \exp\left[\left(\mu - \frac{\sigma^2}{2}\right)dt + \sigma z_t\right]$$

where z_t is a standard Wiener process.

I use Monte Carlo simulation to illustrate how $A(T)$ is distributed. In this notebook, I'll take advantage of *GeometricBrownianMotionProcess*, which was introduced into *Mathematica* in 2012. Simulating using *RandomFunction* is actually quite fast. I make some comments about simulation speed in Appendix C.

In[19]=

```
Module[{GBM, info},
  GBM = GeometricBrownianMotionProcess[μ, σ, S0];
  info = TraditionalForm /@ {Refine[PDF[GBM[t], St], St > 0],
    Mean[GBM[t]], Variance[GBM[t]], CovarianceFunction[GBM, s, t]};
  info = Join[{St1 /@ {"PDF", "mean", "variance", "covariance"}}, {info}];
  Grid[Transpose@info, Frame → All]]
```

Out[19]=


PDF	$\exp\left(-\frac{1}{2\sigma^2 t}\left(\log(S_t) - \log(S_0) - t\left(\mu - \frac{\sigma^2}{2}\right)\right)^2\right) / \left(\sqrt{2\pi} \sigma \sqrt{t} S_t\right)$
mean	$S_0 e^{t\left(\mu - \frac{\sigma^2}{2}\right) + \frac{\sigma^2 t}{2}}$
variance	$\left(e^{\sigma^2 t} - 1\right) e^{2\left(\log(S_0) + t\left(\mu - \frac{\sigma^2}{2}\right) + \sigma^2 t\right)}$
covariance	$S_0^2 e^{\mu(s+t)} \left(e^{\sigma^2 \min(s,t)} - 1\right)$

I generate some representative price paths for nominal parameters (1 year, sampled each month).

In[20]=

```
Module[{S0 = 100, r = 0, q = 0, σ = 0.2, T = 1.0, ΔT = 1/12, pathList, GBM},
  SeedRandom[123];
  GBM = GeometricBrownianMotionProcess[(r - q), σ, S0];
  pathList = RandomFunction[GBM, {0, T, ΔT}, 3];
  ListLinePlot[pathList, AxesLabel → {St1["t"], St1["S"]},
    PlotStyle → {Black, Blue, Green}, PlotMarkers → Automatic]]
```

Out[20]=




Here are some paths together with their corresponding averages.

In[21]=

```
Module[{S0 = 100, r = 0, q = 0, σ = 0.2, T = 1.0,
  ΔT = 1/12, nPaths = 3, pathList, averagePathList, GBM, lab},
  GBM = GeometricBrownianMotionProcess[(r - q), σ, S0];
  SeedRandom[123];
  pathList = Table[RandomFunction[GBM, {0, T, ΔT}], {nPaths}];
  averagePathList = CumulativeAverage /@ pathList;
  (* The TemporalData object returned by RandomFunction is sometimes inconvenient
  The following statement converts it to a simple list of time-value pairs *)
  pathList = Flatten[Normal /@ pathList, 1];
  lab = Stl["price paths and associated cumulative averages"];
  ListLinePlot[Join[pathList, averagePathList], AxesLabel → {Stl["t"], Stl["S"]},
    PlotStyle → {Black, Blue, Green, Directive[Black, Dashed],
      Directive[Blue, Dashed], Directive[Green, Dashed]}, PlotLabel → lab]
```

Out[21]=



In[1]=

```
Clear[CumulativeAverage];
(* This is illustrative. It could be made more elegant or efficient. *)
CumulativeAverage[path_] :=
Module[{dat, times, prices, sums, cumAve},
  dat = Normal[path][[1]];
  {times, prices} = {dat[[All, 1]], dat[[All, 2]]};
  sums = Accumulate[prices];
  cumAve = Table[ $\frac{\text{sums}[[i]]}{i}$ , {i, 1, Length[sums]}];
  Transpose[{times, cumAve}] ]
```

I compare the simulated distributions for $A(T)$ and $S(T)$. Of course, $S(T)$ is lognormal by construction, but $A(T)$ is also well-fitted by that distribution.


In[22]=

```

Module[{S0 = 100, r = 0.05, q = 0, σ = 0.2, T = 1.0, ΔT = 1/12,
  nPaths = 4000, seed = 123, pathList, fST, fAT, DST, DAT, GBM, fittedPDFs },
  GBM = GeometricBrownianMotionProcess[(r - q), σ, S0];
  SeedRandom[seed];
  pathList = Table[RandomFunction[GBM, {0, T, ΔT}], {nPaths}];
  pathList = Flatten[Normal /@ pathList, 1];
  (* ST is the final price in the path *)
  fST = (#[-1, 2]) & /@ pathList;
  (* AT is the mean of the price path *)
  fAT = (Mean[#[[All, 2]]) & /@ pathList;
  (* fitting S(T) and A(T) to lognormal distributions *)
  DST = EstimatedDistribution[fST, LogNormalDistribution[μμ, σσ]];
  DAT = EstimatedDistribution[fAT, LogNormalDistribution[μμ, σσ]];
  fittedPDFs = Module[{Smin, Smax, curve},
    {Smin, Smax} = MinMax[fST];
    curve[1] = ({#, PDF[DST, #]}) & /@ Range[Smin, Smax, 1];
    curve[1] = {Black, Line[curve[1]]};
    curve[2] = ({#, PDF[DAT, #]}) & /@ Range[Smin, Smax, 1];
    curve[2] = {Blue, Line[curve[2]]};
    {curve[1], curve[2]};
  Histogram[{fST, fAT}, 40, "PDF",
  ChartElementFunction → "FadingRectangle", ChartStyle → {Black, Blue},
  PlotLabel → Style["Comparing simulated f(ST) and f(AT) with lognormal fits",
  Bold], Epilog → fittedPDFs]

```

Out[22]=



2 Moment matching for a continuously sampled average

Although discrete sampling of a price path is of the most practical interest, it is useful to derive expressions for the moments of the continuously sampled average

$$A_{\text{cont}}(T) = \frac{1}{T-t} \int_t^T S_u dt \quad (4)$$

If I take $t = 0$, for the sake of simplicity, the first moment is

$$\begin{aligned}
 M_1 &= \mathbb{E}[A_{\text{cont}}(T)] \\
 &= \mathbb{E}\left[\frac{1}{T} \int_0^T S_0 e^{(r - \frac{\sigma^2}{2})u + \sigma z_u} du\right] \\
 &= \frac{S_0}{T} \frac{e^{rT} - 1}{r}
 \end{aligned} \tag{5}$$

This result is derived in detail in Appendix A.

I define a function for this moment.

In[23]:=

```
Clear[M1Continuous];
M1Continuous[S0_, r_, T_] :=
  ( -1 + e^{r T} ) S0
  / ( r T )
```

Similarly, the second moment is

$$\begin{aligned}
 M_2 &= \mathbb{E}[A_{\text{cont}}(T)^2] \\
 &= \left(\frac{S_0}{T}\right)^2 \mathbb{E}\left[\int_0^T \int_0^T e^{(r - \frac{\sigma^2}{2})u + \sigma z_u} e^{(r - \frac{\sigma^2}{2})v + \sigma z_v} dv du\right] \\
 &= \left(\frac{S_0}{T}\right)^2 \left(\left(2 \left((1 - 2e^{rT} + e^{T(2r + \sigma^2)})r - (-1 + e^{rT})\sigma^2 \right) \right) / (r(r + \sigma^2)(2r + \sigma^2)) \right)
 \end{aligned} \tag{6}$$

In[25]:=

```
Clear[M2Continuous];
M2Continuous[S0_, r_, sigma_, T_] :=
  ( (S0/T)^2 ) ( ( 2 ( (1 - 2 e^{r T} + e^{T(2 r + sigma^2)}) r - (-1 + e^{r T}) sigma^2 ) ) / ( r ( r + sigma^2 ) ( 2 r + sigma^2 ) ) )
```

The moments of a lognormal distribution with parameters μA and σA are

In[27]:=

```
w2[1] = Module[{D},
  D = LogNormalDistribution[μA, σA];
  {M1 == Moment[D, 1], M2 == Moment[D, 2]}]
```

Out[27]=

```
{M1 == e^{μA + σA^2/2}, M2 == e^{2 μA + 2 σA^2}}
```

In[28]=

```
w2[2] = Solve[w2[1], {μA, σA}] /. {C[1] → 0, C[2] → -0}
```

Out[28]=

$$\left\{ \left\{ \mu_A \rightarrow \text{Log} \left[-\frac{M1^2}{\sqrt{M2}} \right], \sigma_A \rightarrow -\sqrt{2} \sqrt{\text{Log} \left[-\frac{\sqrt{M2}}{M1} \right]} \right\}, \right.$$

$$\left. \left\{ \mu_A \rightarrow \text{Log} \left[-\frac{M1^2}{\sqrt{M2}} \right], \sigma_A \rightarrow \sqrt{2} \sqrt{\text{Log} \left[-\frac{\sqrt{M2}}{M1} \right]} \right\}, \right.$$

$$\left. \left\{ \mu_A \rightarrow \text{Log} \left[\frac{M1^2}{\sqrt{M2}} \right], \sigma_A \rightarrow -\sqrt{2} \sqrt{\text{Log} \left[\frac{\sqrt{M2}}{M1} \right]} \right\}, \left\{ \mu_A \rightarrow \text{Log} \left[\frac{M1^2}{\sqrt{M2}} \right], \sigma_A \rightarrow \sqrt{2} \sqrt{\text{Log} \left[\frac{\sqrt{M2}}{M1} \right]} \right\} \right\}$$

The financially relevant branch is the fourth one

In[29]=

```
w2[3] = w2[2][[4]]
```

Out[29]=

$$\left\{ \mu_A \rightarrow \text{Log} \left[\frac{M1^2}{\sqrt{M2}} \right], \sigma_A \rightarrow \sqrt{2} \sqrt{\text{Log} \left[\frac{\sqrt{M2}}{M1} \right]} \right\}$$

I need to relate the parameters of this distribution to the calculated moments.

In[34]=

```
w2[4] = {M1 == Moment[LogNormalDistribution[μA, σA], 1],  
M2 == Moment[LogNormalDistribution[μA, σA], 2]} // Simplify
```

Out[34]=

$$\left\{ e^{\mu_A + \frac{\sigma_A^2}{2}} == M1, e^{2(\mu_A + \sigma_A^2)} == M2 \right\}$$

For example,


In[36]=

```

Module[{S0 = 100, r = 0.01, q = 0, σ = 0.2, T = 1.0, μA, σA, M1, M2, D},
  M1 = M1Continuous[S0, r, T];
  M2 = M2Continuous[S0, r, σ, T];
  μA =  $\frac{1}{2} (4 \text{Log}[M1] - \text{Log}[M2])$ ;
  σA =  $\sqrt{-2 \text{Log}[M1] + \text{Log}[M2]}$ ;
  D = LogNormalDistribution[μA, σA];
  Plot[PDF[D, A], {A, 50, 150},
    PlotStyle → Directive[Thick, Black],
    PlotLabel → St1["Moment matched log normal distribution -- continuous"],
    AxesLabel → {St1["A"], St1["f(A)"]}]]

```

Out[36]=



I check this moment matched distribution by comparing against a simulated distribution for monthly sampling.

In[37]=

```


Module[{S0 = 100, r = 0.01, q = 0, σ = 0.2, T = 1.0, ΔT = 1/12, nPaths = 4000,
  seed = 123, pathList, fAT, H, M1, M2, μA, σA, D, GBM, curve, G},
  GBM = GeometricBrownianMotionProcess[(r - q), σ, S0];
  SeedRandom[seed];
  pathList = Table[RandomFunction[GBM, {0, T, ΔT}], {nPaths}];
  pathList = Flatten[Normal /@ pathList, 1];
  (* The monthly sampled simulated A(T) *)
  fAT = (Mean[#[[All, 2]]] & /@ pathList;
  H = HistogramDistribution[fAT];

  (* The continuously sampled case for the same parameters *)
  M1 = M1Continuous[S0, r, T];
  M2 = M2Continuous[S0, r, σ, T];
  μA =  $\frac{1}{2} (4 \text{Log}[M1] - \text{Log}[M2])$ ;
  σA =  $\sqrt{-2 \text{Log}[M1] + \text{Log}[M2]}$ ;
  D = LogNormalDistribution[μA, σA];

  G[1] = DiscretePlot[PDF[H, A],
    {A, Min[fAT], Max[fAT]}, ExtentSize → Full, PlotStyle → Blue,
    PlotLabel → St1["discretely sampled simulated distribution (blue)\nmoment matched
      continuous sampling (black)"], AxesLabel → {St1["AT"], St1["f(AT)"]};
  G[2] = Plot[PDF[D, A], {A, Min[fAT], Max[fAT]}];
  Show[{G[1], G[2]}]

```

Out[37]=



Agreement is remarkable, considering there were only 12 monthly samples per price path.

3 Moment matching for a discretely sampled average

I now consider the case of a discrete average with sample time $\Delta t = \frac{T}{n}$. See Appendix B for details of the moment calculation.

$$\begin{aligned} M1 &= \mathbb{E}[A] \\ &= \mathbb{E}\left[\frac{1}{n+1} \sum_{i=0}^n S_i\right] \\ &= \frac{S_0}{(1+n)} \frac{(-1 + e^{r\Delta t + nr\Delta t})}{(-1 + e^{r\Delta t})} \end{aligned}$$

In[38]:=

```
Clear[M1Discrete];
M1Discrete[S0_, r_, dt_, n_] :=  $\frac{S0}{n+1} \frac{(\text{Exp}[(n+1)r dt] - 1)}{(\text{Exp}[r dt] - 1)}$ 
```

The discrete second moment is

$$\begin{aligned} \mathbb{E}[A^2] &= \mathbb{E}\left[\left(\frac{1}{n+1} \sum_{i=0}^n S_i\right)\left(\frac{1}{n+1} \sum_{j=0}^n S_j\right)\right] \\ &= \left(\frac{(-1-x+2x^{1+n}+xy+x^2y-2x^{3+n}y-x^{2+2n}y^{1+n}+x^{3+2n}y^{1+n}-x^{3+2n}y^{2+n}+x^{4+2n}y^{2+n})S_0^2}{(1+n)^2(-1+x)(-1+xy)(-1+x^2y)}\right) \end{aligned}$$

where $x = e^{rt}$ and $y = e^{\sigma^2 \Delta t}$. See Appendix B for the calculation

In[40]:=


```
Clear[M2Discrete];
M2Discrete[S0_, r_, sigma_, T_, n_] :=
Module[{x, y, dt =  $\frac{T}{n}$ , doubleSum, M2},
  x = Exp[r dt];
  y = Exp[sigma^2 dt];
  doubleSum =
     $\frac{(-1-x+2x^{1+n}+xy+x^2y-2x^{3+n}y-x^{2+2n}y^{1+n}+x^{3+2n}y^{1+n}-x^{3+2n}y^{2+n}+x^{4+2n}y^{2+n})}{((-1+x)(-1+xy)(-1+x^2y))}$ ;
  M2 =  $\left(\frac{S0}{n+1}\right)^2$  doubleSum]
```

I compare the discrete sampling result with continuous sampling.

In[43]=

```
Module[{S0 = 100, r = 0.1, σ = 0.2, T = 1, nDoub = 8, vals, M2Cont},
  M2Cont = M2Continuous[S0, r, σ, T];
  vals = Table[{Log[2i], M2Discrete[S0, r, σ, T, 2i]}, {i, 1, nDoub}];
  Graphics[{Line[vals], {PointSize[0.01], Point /@ vals},
    {Directive[Thick, Red], Line[{Log[2], M2Cont}, {Log[2nDoub], M2Cont}]}}],
  Axes → Automatic, AspectRatio → 0.5,
  PlotLabel → St1["Comparing M2 for discrete and continuous sampling"],
  AxesLabel → {Style["Log[2i]", Bold], Style["M2(Δt = T/2i)", Bold]}]
```

Out[43]=



The black line represents M_2 for discrete sampling. The sampling interval is $\Delta t = \frac{T}{n}$, with $n = 2^i$. The abscissa is scaled logarithmically so that 8 doublings of the sample frequency can be accommodated. The red line represents M_2 for continuous sampling. For a sampling frequency of $2^8 = 256$, the discrete value for M_2 is quite close to the result for continuous sampling.

Appendix A - Moment calculations for continuously sampled average

First moment calculation: The symbols \mathbb{E} (for expectation) and Int (for integration) are abstract operators that I manipulate with rules

In[44]=

$$m1[1] = \mathbb{E} \left[\frac{1}{T} \text{Int}[A[u], \{u, \theta, T\}] \right] /. A[u] \rightarrow S_0 \text{Exp} \left[\left(r - \frac{\sigma^2}{2} \right) u + \sigma z[u] \right]$$

Out[44]=

$$\mathbb{E} \left[\frac{\text{Int} \left[e^{u \left(r - \frac{\sigma^2}{2} \right) + \sigma z[u]} S_0, \{u, \theta, T\} \right]}{T} \right]$$

Interchange expectation and integration operators

In[45]=

$$m1[2] = m1[1] /. \mathbb{E} [a_. \text{Int}[b_, c_]] \rightarrow a \text{Int}[\mathbb{E}[b], c]$$

Out[45]=

$$\frac{1}{T} \text{Int} \left[\mathbb{E} \left[e^{u \left(r - \frac{\sigma^2}{2} \right) + \sigma z[u]} S_0 \right], \{u, \theta, T\} \right]$$

Make the expectation explicit

In[46]:=

m1[3] = m1[2] /. $\mathbb{E}[a_]$ \rightarrow Expectation[a, z \approx WienerProcess[0, 1]]

Out[46]=

$$\frac{\text{Int}[e^{r u} S_0, \{u, \theta, T\}]}{T}$$

Perform the integral

In[47]:=

m1[4] = m1[3] /. Int \rightarrow Integrate

Out[47]=

$$\frac{(-1 + e^{r T}) S_0}{r T}$$

Second moment calculation

In[48]:=

m2[1] = $\mathbb{E}\left[\frac{1}{T^2} \text{Int}[A[u] A[v], \{u, \theta, T\}, \{v, \theta, T\}]\right] /. A[x_] \rightarrow S_0 \text{Exp}\left[\left(r - \frac{\sigma^2}{2}\right) x + \sigma z[x]\right]$

Out[48]=

$$\mathbb{E}\left[\frac{1}{T^2} \text{Int}\left[e^{u\left(r - \frac{\sigma^2}{2}\right) + v\left(r - \frac{\sigma^2}{2}\right) + \sigma z[u] + \sigma z[v]} S_0^2, \{u, \theta, T\}, \{v, \theta, T\}\right]\right]$$

In[49]:=

m2[2] = m2[1] /. $\mathbb{E}[a_. \text{Int}[b_, c_, d_]] \rightarrow a \text{Int}[\mathbb{E}[b], c, d]$ // Simplify

Out[49]=

$$\frac{1}{T^2} \text{Int}\left[\mathbb{E}\left[e^{\frac{1}{2}(u+v)(2r - \sigma^2) + \sigma z[u] + \sigma z[v]} S_0^2\right], \{u, \theta, T\}, \{v, \theta, T\}\right]$$

The expectation calculation is simplified if I invoke an identity

In[50]:=

m2[3] = m2[2] /. $\mathbb{E}[c_. \text{Exp}[a_. + b_]] /; \text{FreeQ}[a, z] \rightarrow c \text{Exp}[a] \text{Exp}\left[\frac{\mathbb{E}[b^2]}{2}\right]$ // ExpandAll

Out[50]=

$$\frac{1}{T^2} \text{Int}\left[e^{r u + r v - \frac{u \sigma^2}{2} - \frac{v \sigma^2}{2} + \frac{1}{2} \mathbb{E}[\sigma^2 z[u]^2 + 2 \sigma^2 z[u] z[v] + \sigma^2 z[v]^2]} S_0^2, \{u, \theta, T\}, \{v, \theta, T\}\right]$$

Distribute the expectation

In[51]:=

m2[4] = m2[3] //. $\mathbb{E}[a_ + b_] \rightarrow \mathbb{E}[a] + \mathbb{E}[b]$

Out[51]=

$$\frac{1}{T^2} \text{Int}\left[e^{r u + r v - \frac{u \sigma^2}{2} - \frac{v \sigma^2}{2} + \frac{1}{2} (\mathbb{E}[\sigma^2 z[u]^2] + \mathbb{E}[2 \sigma^2 z[u] z[v]] + \mathbb{E}[\sigma^2 z[v]^2])} S_0^2, \{u, \theta, T\}, \{v, \theta, T\}\right]$$

Perform the expectation

In[52]:= $m2[5] = m2[4] /. \mathbb{E}[a_z[x_]^2] \rightarrow \text{Expectation}[a_z[x]^2, z \approx \text{WienerProcess}[0, 1]] /.$
 $\mathbb{E}[a_z[x_] z[y_]] \rightarrow a \text{CovarianceFunction}[\text{WienerProcess}[0, 1], x, y] // \text{Simplify}$

Out[52]=
$$\frac{1}{T^2} \text{Int} \left[e^{r(u+v) + \sigma^2 \text{Min}[u,v]} S_0^2, \{u, 0, T\}, \{v, 0, T\} \right]$$

Mathematica takes a loong time to value such an integral. I speed things up by spitting the v integration into two parts $0 < v < u$ and $u < v < T$

In[53]:= $\text{part}[1] = (m2[5] /. \text{Int}[a_, \{u, 0, T\}, \{v, 0, T\}] \rightarrow \text{Integrate}[(a /. \text{Min}[u, v] \rightarrow v), \{v, 0, u\}])$

Out[53]=
$$\frac{e^{ru} (-1 + e^{u(r+\sigma^2)}) S_0^2}{T^2 (r + \sigma^2)}$$

In[54]:= $\text{part}[2] = (m2[5] /. \text{Int}[a_, \{u, 0, T\}, \{v, 0, T\}] \rightarrow \text{Integrate}[(a /. \text{Min}[u, v] \rightarrow u), \{v, u, T\}])$

Out[54]=
$$\frac{e^{u\sigma^2} (-e^{2ru} + e^{r(T+u)}) S_0^2}{r T^2}$$

Finally,

In[55]:= $m2[6] = \text{Integrate}[\text{part}[1] + \text{part}[2], \{u, 0, T\}]$

Out[55]=
$$\left(2 \left((1 - 2e^{rT} + e^{T(2r+\sigma^2)}) r - (-1 + e^{rT}) \sigma^2 \right) S_0^2 \right) / (r T^2 (r + \sigma^2) (2r + \sigma^2))$$

Appendix B - Moment calculations for discretely sampled average

$$\begin{aligned} M1 &= \mathbb{E}[A] \\ &= \mathbb{E}\left[\frac{1}{n+1} \sum_{i=0}^n S_i\right] \\ &= \frac{S_0}{n} \frac{(e^{(n+1)r\Delta t} - 1)}{(e^{r\Delta t} - 1)} \end{aligned}$$

Define the expression

$$\text{In[58]:= } m1B[1] = \mathbb{E} \left[\frac{1}{n+1} \text{Sm}[S_i, \{i, \theta, n\}] \right] /. S_i \rightarrow S_\theta \text{Exp} \left[\left(r - \frac{\sigma^2}{2} \right) i \Delta t + \sigma z[i \Delta t] \right]$$

$$\text{Out[58]:= } \mathbb{E} \left[\frac{1}{1+n} \text{Sm} \left[e^{i \Delta t \left(r - \frac{\sigma^2}{2} \right) + \sigma z[i \Delta t]} S_\theta, \{i, \theta, n\} \right] \right]$$

Interchange expectation and summation

$$\text{In[59]:= } m1B[2] = m1B[1] /. \mathbb{E}[a_. \text{Sm}[b_, c_]] \rightarrow a \text{Sm}[\mathbb{E}[b], c]$$

$$\text{Out[59]:= } \frac{1}{1+n} \text{Sm} \left[\mathbb{E} \left[e^{i \Delta t \left(r - \frac{\sigma^2}{2} \right) + \sigma z[i \Delta t]} S_\theta \right], \{i, \theta, n\} \right]$$

Make the expectation explicit

$$\text{In[60]:= } m1B[3] = m1B[2] /. \mathbb{E}[a_] \Rightarrow \text{Expectation}[a, z \approx \text{WienerProcess}[\theta, 1]]$$

$$\text{Out[60]:= } \frac{\text{Sm} \left[e^{i r \Delta t} S_\theta, \{i, \theta, n\} \right]}{1+n}$$

Evaluate the sum

$$\text{In[61]:= } m1B[4] = m1B[3] /. \text{Sm} \rightarrow \text{Sum}$$

$$\text{Out[61]:= } \frac{(-1 + e^{r \Delta t + n r \Delta t}) S_\theta}{(-1 + e^{r \Delta t}) (1+n)}$$

Second moment calculation.

$$\text{In[62]:= } m2B[1] = \mathbb{E} \left[\left(\text{Sm} \left[\frac{S_i}{n+1} \frac{S_j}{n+1}, \{i, \theta, n\}, \{j, \theta, n\} \right] \right)^2 \right]$$

$$\text{Out[62]:= } \mathbb{E} \left[\text{Sm} \left[\frac{S_i S_j}{(1+n)^2}, \{i, \theta, n\}, \{j, \theta, n\} \right] \right]$$

$$\text{In[63]:= } m2B[2] = m2B[1] /. S_{i_} \rightarrow S_\theta \text{Exp} \left[\left(r - \frac{\sigma^2}{2} \right) i \Delta t + \sigma z[i \Delta t] \right]$$

$$\text{Out[63]:= } \mathbb{E} \left[\text{Sm} \left[\frac{1}{(1+n)^2} e^{i \Delta t \left(r - \frac{\sigma^2}{2} \right) + j \Delta t \left(r - \frac{\sigma^2}{2} \right) + \sigma z[i \Delta t] + \sigma z[j \Delta t]} S_\theta^2, \{i, \theta, n\}, \{j, \theta, n\} \right] \right]$$

$$\text{In[64]:= } m2B[3] = m2B[2] /. \mathbb{E}[\text{Sm}[a_, b_, c_]] \rightarrow \text{Sm}[\mathbb{E}[a], b, c]$$

$$\text{Out[64]:= } \text{Sm} \left[\mathbb{E} \left[\frac{1}{(1+n)^2} e^{i \Delta t \left(r - \frac{\sigma^2}{2} \right) + j \Delta t \left(r - \frac{\sigma^2}{2} \right) + \sigma z[i \Delta t] + \sigma z[j \Delta t]} S_\theta^2 \right], \{i, \theta, n\}, \{j, \theta, n\} \right]$$

In[65]:= **m2B[4] = m2B[3] /. $\mathbb{E}[a_ \text{Exp}[b_ + c_]]$ /; FreeQ[c, z] \rightarrow a Exp[c] $\mathbb{E}[\text{Exp}[b]]$**

Out[65]=
$$\text{Sm} \left[\frac{1}{(1+n)^2} e^{i \Delta t \left(r - \frac{\sigma^2}{2}\right) + j \Delta t \left(r - \frac{\sigma^2}{2}\right)} S_0^2 \mathbb{E} \left[e^{\sigma z [i \Delta t] + \sigma z [j \Delta t]} \right], \{i, \theta, n\}, \{j, \theta, n\} \right]$$

In[66]:= **m2B[5] = m2B[4] /. $\mathbb{E}[\text{Exp}[a_]] \rightarrow \text{Exp}[\mathbb{E}\left[\frac{a^2}{2}\right]]$ // ExpandAll**

Out[66]=
$$\text{Sm} \left[\left(e^{i r \Delta t + j r \Delta t - \frac{1}{2} i \Delta t \sigma^2 - \frac{1}{2} j \Delta t \sigma^2 + \mathbb{E} \left[\frac{1}{2} \sigma^2 z [i \Delta t]^2 + \sigma^2 z [i \Delta t] z [j \Delta t] + \frac{1}{2} \sigma^2 z [j \Delta t]^2 \right]} S_0^2 \right) / (1 + 2n + n^2), \{i, \theta, n\}, \{j, \theta, n\} \right]$$

In[67]:= **m2B[6] = m2B[5] //. $\mathbb{E}[a_ + b_]$ \rightarrow $\mathbb{E}[a] + \mathbb{E}[b]$**

Out[67]=
$$\text{Sm} \left[\left(e^{i r \Delta t + j r \Delta t - \frac{1}{2} i \Delta t \sigma^2 - \frac{1}{2} j \Delta t \sigma^2 + \mathbb{E} \left[\frac{1}{2} \sigma^2 z [i \Delta t]^2 \right] + \mathbb{E} \left[\sigma^2 z [i \Delta t] z [j \Delta t] \right] + \mathbb{E} \left[\frac{1}{2} \sigma^2 z [j \Delta t]^2 \right]} S_0^2 \right) / (1 + 2n + n^2), \{i, \theta, n\}, \{j, \theta, n\} \right]$$

In[68]:= **m2B[7] = m2B[6] /. $\mathbb{E}[a_ z [i_ \Delta t]^2]$ \rightarrow Expectation[a z [i Δt]², z \approx WienerProcess[0, 1]] /. $\mathbb{E}[a_ z [x_]$ z [y_]] \rightarrow a CovarianceFunction[WienerProcess[0, 1], x, y] // Simplify**

Out[68]=
$$\text{Sm} \left[\frac{e^{(i+j) r \Delta t + \sigma^2 \text{Min}[i \Delta t, j \Delta t]} S_0^2}{(1+n)^2}, \{i, \theta, n\}, \{j, \theta, n\} \right]$$

It's worthwhile to simplify the notation before performing the integration

In[69]:= **rRule = Sol[x == Exp[r Δt], r] /. C[1] \rightarrow 0**

Out[69]=
$$r \rightarrow \frac{\text{Log}[x]}{\Delta t}$$

In[70]:= **σ Rule = Solve[y == Exp[$\sigma^2 \Delta t$], σ] [[2, 1]] /. C[1] \rightarrow 0**

Out[70]=
$$\sigma \rightarrow \frac{\sqrt{\text{Log}[y]}}{\sqrt{\Delta t}}$$

In[71]:= **m2B[8] = m2B[7] /. Min[i Δt , j Δt] \rightarrow Δt Min[i, j] /. rRule /. σ Rule // Simplify**

Out[71]=
$$\text{Sm} \left[\frac{x^{i+j} y^{\text{Min}[i, j]} S_0^2}{(1+n)^2}, \{i, \theta, n\}, \{j, \theta, n\} \right]$$

For some fun, try calculating this sum by hand. *Pricing european average rate currency options*, Edward Levy (1992). In the late 90's it took a couple of days to establish confidence in Levy's results. Mathematica calculates the sum in a few seconds.

In[72]:=

m2B[9] = m2B[8] /. Sm → Sum

$$\left[\begin{array}{l} \frac{S_0^2}{(1+n)^2} \\ \left((-1-x+2x^{1+n}+xy+x^2y-2x^{3+n}y-x^{2+2n}y^{1+n}+x^{3+2n}y^{1+n}-x^{3+2n}y^{2+n}+x^{4+2n}y^{2+n}) S_0^2 \right) / \text{True} \\ \left((1+n)^2 (-1+x) (-1+xy) (-1+x^2y) \right) \end{array} \right] \quad n == 0$$

In[73]:=

m2B[10] = Refine[m2B[9], n > 0]

$$\left((-1-x+2x^{1+n}+xy+x^2y-2x^{3+n}y-x^{2+2n}y^{1+n}+x^{3+2n}y^{1+n}-x^{3+2n}y^{2+n}+x^{4+2n}y^{2+n}) S_0^2 \right) / \left((1+n)^2 (-1+x) (-1+xy) (-1+x^2y) \right)$$

Appendix C Simulation speed

Just out of curiosity, I perform a simple speed test that compares simulation using RandomFunction of GeometricBrownianMotionProcess to methods I have used in the past. RandomFunction actually seems to be faster than the path generators I used in the past.

In[74]:=

```
Module[{S0 = 100, r = 0.02, q = 0, σ = 0.2,
  T = 1.0, ΔT = 1/52, nPaths = 100000, pathList, GBM},
  SeedRandom[123];
  GBM = GeometricBrownianMotionProcess[(r - q), σ, S0];
  pathList = Timing[RandomFunction[GBM, {0, T, ΔT}, nPaths]];
  pathList[[1]]]
```

Out[74]=

0.1875

In[75]:=

```
GBMPathCompiled =
  Compile[{{S0, _Real}, {drift, _Real}, {diff, _Real}, {nSteps, _Integer}},
  FoldList[ (#1 drift Exp[diff #2]) &, S0,
  RandomVariate[NormalDistribution[0, 1], nSteps] ]];
```

In[76]:=

```
Module[{S0 = 100, r = 0.02, q = 0, σ = 0.2, T = 1.0,
  ΔT = 1/52, nSteps = 52, nPaths = 100000, drift, diff, pathList, GBM},
  SeedRandom[123];
  drift = Exp[ (r - q - (σ^2)/2) ΔT ];
  diff = σ Sqrt[ΔT];
  pathList = Timing[Table[GBMPathCompiled[S0, drift, diff, nSteps], {nPaths}]];
  pathList[[1]]]
```

Out[76]=

0.46875

In[15]=

```
Clear[GBMPath];
GBMPath[S0_, drift_, diff_, nSteps_] :=
  FoldList[ (#1 drift Exp[ diff #2]) &,
    S0, RandomVariate[NormalDistribution[0, 1], nSteps]
```

In[77]=

```
Module[{S0 = 100, r = 0.02, q = 0, σ = 0.2, T = 1.0,
  ΔT = 1/52, nSteps = 52, nPaths = 100000, drift, diff, pathList, GBM},
  SeedRandom[123];
  drift = Exp[ (r - q -  $\frac{\sigma^2}{2}$ ) ΔT];
  diff = σ √ΔT;
  pathList = Timing[Table[GBMPath[S0, drift, diff, nSteps], {nPaths}]];
  pathList [[1]] ]
```

Out[77]=

6.375